

קלט / פלט

עד כה עשינו קלט / פלט דרך ה-BIOS (ישירות או בעקיפין דרך DOS). למען האמת העברנו בקשות קלט / פלט + אינפורמציה לתוכניות אסמבלי שנכתבו ע"י היצרן. כל תוכנית אסמבלי יכול לכתוב את הרוטינות הללו, אם כי הרוטינות עשויות להיות תלויות בפלטפורמה הספציפית שבה מדובר, למרות שבפירוש ישנסיון לממש סטנדרטיזציה גם כאן. תוכנית ה-keyboard המובאת להלן, keyb6.asm, היא תוכנית שלקחה על עצמה את תפקידי רוטינות ה-BIOS, והתנהגות שלה היתה מעט שונה על מקלדות שונות. ביחוד בכל הקשור למקשי החיצים, Home, Ins וכו'. בטרמינולוגיה מקצועית אפשר לומר שהתוכנית מכילה בתוכה Device Driver של המקלדת.

שיטות קלט / פלט

התיעוד של חברת Intel קובע שיש שתי שיטות קלט / פלט עיקריות:

- קלט / פלט דרך ports (ערוצים?)

- Memory Mapped I/O (קלט / פלט ממופה לכתובות).

לדעתי אפשר להוסיף קטגוריה שלישית: Direct Memory Access (DMA). אנשי Intel כנראה קטלגו שיטה זו במסגרת הקטגוריה של ports.

ports הם מעין מרחב כתובות (נפרדת מהזיכרון) המיצגים (בעיני התוכניתן לפחות) מעין חוטים עם 2 קצוות: קצה ב-CPU וקצה שני בהתקן ק/פ (בדרך כלל chip אחר). עד כמה שהבנתי, יתכן שלפחות היסטורית זה היה פעם גם המימוש הפיזי - יתכן שחלק מהקוים הכסופים על ה-mother board הינם (או היו) מימושים של ports. השאלה איך מתבטא ports בחומרה הוא חסר חשיבות לתוכניתן - הדבר מן הסתם השתנה בין גירסת PC אחד למשנהו, ובין יצרן ליצרן. מרחב הכתובות של ה-ports הוא בין 0 ל-65535 (רובם לא בשימוש).

בכל מקרה אם port מחובר למשהו או לא - זה עניין של חומרה - אי אפשר בתוכנה "לחבר" port למשהו.

יש לשים לב: מספרי port הם חיבורים ישירים וגלויים ל-CPU. ישנם חיבורים במחשב בין chips שאינם ה-CPU והם אינם נכללים בקטגוריה הזו, שכן אינם

גלויים ל-CPU (או, אם תרצו, לתוכניתן האסמבלי).

Memory Mapped I/O הוא מציאות שבה כתובות זיכרון מסוימות מנותות להחקני ק/פ במקום לזיכרון. הגישה הזו מחייבת מימוש בחומרה. שום תוכנה אינה יכולה "למש" Memory Mapped I/O בכוחות עצמה. היצרן בונה את המחשב כך שכתובה / קריאה לכתובות מסוימים משמעותם למעשה ק/פ. הדוגמא הקלאסית היא כתיבה / קריאה לזיכרון המסך: הסגמנטים A000h (כתובת פיזית 640k), B000h (כתובת פיזית 704k), B800h (כתובת פיזית 736k) הם כתובות המנותות לזיכרון של כרטיס המסך של המחשב. תוכנית הכותבת / קוראת ל"שטחי זיכרון" הללו למעשה כותבת / קוראת לזיכרון של כרטיס המסך, והדבר מתבטא בשינוי התצוגה של המסך.

DMA הוא מאין שילוב של שני הקודמים: דרך ports ה-CPU מנחה chip של DMA (Intel 8237A) לקרוא ממקום מסוים בהתקן (נניח Track מסוים בדיסק) ליעד מסוים בזיכרון (או להיפך). המידע אינו עובר דרך ה-CPU, דבר שהיה אמור לחסוך overhead ולאפשר ל-CPU לעסוק בביכול בדברים אחרים (במקביל ל-DMA).

פקודות מכונה לקריאה / כתיבה ל-ports:

יש מספר כאילו, אבל הם למען האמת גירסאות של הפקודות IN, OUT.

מבנה הפקודה IN:

AL	port מספר
IN	או , או
AX	DX
או	
EAX (386 ואילך)	

קורא בית או מילה או מילה כפולה מ-port לאוגר האקומולטור (AL, AX, EAX). על משמעות של קריאה של יותר מבית אחד נראה בהמשך.

מספר port הוא קבוע בין 0 ל-255.

מבנה הפקודה OUT:

AL
מספר port
או , או
AX
DX
או
EAX (386 ואילך)

כותב בית או מילה או מילה כפולה מהאקומולטור (AL, AX, EAX) ל-
port. על משמעות של כתיבה של יותר מבית אחד נראה בהמשך.

מספר port הוא קבוע בין 0 ל-255.

דוגמאות:

IN AL,60h

קריאת תו מהמקלדת

MOV DX,378h
OUT DX,AL

שליחת תו למדפסת.

מספר port המופיע במפורש בפקודות IN או OUT חייב להיות קבוע בין 0 ל-255.
אפשר לגשת לכל מספר port דרך הגירסאות של הפקודות המשתמשות בתוכן DX
לציון מספר ה-port. אולם ל-port עם מספר גדול יותר מ-255 אפשר לגשת רק
ע"י גירסאות ה-DX של הפקודות.
מספר ה-port המירבי ב-DX הוא 65535. אין מספרי port גדולים יותר (אין
תמיכה בגירסה של הפקודה עם EDI, גם לא בפנטיום).

הגדלים של האופרנדים אינם בהכרח שווים כמובן (שכן מספר ה-port או DX הם
מעין פוינטרים).

כאשר משתמשים ב-AX או EAX בפקודות IN ו-OUT, ה-bytes המשמעותיים יותר של
האוגר נכתבים למספרי ports עוקבים, כאילו היו כתובות. לדוגמא, אם מתבצע:

OUT 200,AX

אז AL נכתב לתוך port מספר 200, AH נכתב לתוך port מספר 201.

אי אפשר סתם לשלוח תוים ל-port.

יש צורך לעקוב אחרי פרוטוקול של בדיקות סטטוס ואישרור קבלה.

רוב מספרי ה-port אינם מנוצלים, ואינם מחוברים לדבר. שום תקלה אינה נגרמת כאשר קוראים זבל מ-port שאינו מחבר לכלום או כותבים ל-port שאינו מחבר לכלום.

Support Chips

Support Chips הם מעין מיקרו פרוססורים קטנים ל"משימות מיוחדות". הם מורידים משימות מה-CPU על מנת שלפשוט את מימוש ה-CPU. הם פשוטים יותר מה-CPU ובדרך כלל ניתנים "לתכנות" בצורה מוגבלת - אפשר לשנות את התנהגותם ע"י כתיבה לאוגרים שלהם דרך ports. דוגמא ל-Support Chip הוא ה-Timer ששולח פולס ל-CPU 18.207 פעמים בשניה, ומאפשר ל-CPU לממש את פסיקה מספר 8. החסרון של שימוש ב-Support Chips הוא שהדבר מאיט את המערכת ככלל.

נתרכז בשני Support Chips אחרים: בקר הפסיקות ובקר ההתקנים החיצוניים.

דוגמא לקריאת התקן דרך ports: המקלדת (פסיקה מספר 9).

פסיקה מספר 9 היא פסיקת החומרה של המקלדת. המערכת של המחשב בנויה כך שכל לחיצה / שחרור של מקש במקלדת גורם לפסיקת חומרה 9. הקריאה של איזה מקש מדובר נעשה דרך ports. למעשה תפקידה של פסיקה 9 לשמש מנגנון העברת אינפורמציה של המקלדת לתוכנה. ה-ISR של 9 מעביר את תוכן המקש לזיכרון של ה-BIOS, ופסיקת התוכנה 16h לוקחת אותו משם.

איך בדיוק הרעיון הזה מיושם תלוי מאד באיזה דגם של מחשב מדובר. התאור הבא מתאים ל-8086. בדגמים שבאו אחר כך חלו שינויים, לא נעמוד עליהם כאן.

התהליך של העברת מידע מהמקלדת הוא הבא:

בתוך המקלדת יש מיקרו-פרוססור קטן המבחין בכל לחיצה ובכל שחרור של מקש. הוא מאותת דרך חיבור מיוחד למחשב שיש לו מידע למחשב (החיבור הזה אינו גלוי ל-CPU, לכן אין לו מספר port). החיבור הזה הוא ל-chip שמשמש בקר התקנים חיצוניים. המקלדת אינה, כמובן, ההתקן היחיד שמחובר לבקר הזה.

בקר ההתקנים מחובר ל-CPU (בין השאר) ע"י ה-ports 60h, ו-61h.

port מספר 60h משמש להעברת ה-scan code, שבקר ההתקנים מקבל מהמקלדת ל-CPU. ה-scan code הוא מספר קטן או שווה ל-127. הביט המשמעותי ביותר ב-port הוא איפוא פנוי והוא משמש להבדיל בין לחיצה לשחרור. 0 = לחיצה, 1 = שחרור.

port מספר 61h משמש את ה-CPU "להודיע" לבקר ההתקנים שהתו הנוכחי נקרא ע"י ה-CPU.

ה-port הזה הוא דו-סיטרי: בקר ההתקנים מעביר קוד בין 0 ל-127 (למעשה מספר סידורי) ל-CPU, ה-CPU מעביר אותו בחזרה לבקר ההתקנים תוך הדלקת הביט המשמעותי ביותר, ואחר כך שוב את אותו הקוד עם הביט המשמעותי ביותר כבוי.

אחרי ההודעה מהמקלדת, בקר ההתקנים מבקש מבקר הפסיקות לגרום לפסיקה מספר 9 ב-CPU. ה-CPU מחובר לבקר הפסיקות (בין השאר) דרך port 20h. כחלק מהטיפול בפסיקה 9 (או כל פסיקת חומרה אחרת) על ה-CPU לאותת "סיום טיפול בפסיקה" ע"י כתיבת הערך 20h לתוך port 20h. השיוון בין מספר ה-port והתוכן הוא, עד כמה שידוע לי, מקרי.

לפיכך התהליך של טיפול בפסיקה 9 ע"י ה-ISR, כאשר הסדר מחייב, הוא כלהלן:

א. קרא את התו דרך port 60h

ב. הודע לבקר ההתקנים שהתו נקרא, דרך port 61h.


הדרך לעשות זאת הינו לקרוא את תוכן port 61h, להדליק את הביט העליון של המספר 8-ביט הזה, לשגר אותו חזרה ל-port 61h, לכבות את הביט העליון, ולשגר שוב את המספר עם הביט העליון כבוי. ראה קטע קוד III בתוכנית.

ג. הודע לבקר הפסיקות שהפסיקה טופלה, דרך port 20h.

הדרך לעשות זאת הינו לשגר את המספר 20h לתוך port 20h. ראה קטע קוד V בתוכנית.

ב-8086 שימש ה-Intel 8259A בתור בקר הפסיקות, וה-Intel 8255 בתור בקר ההתקנים. במחשבים שכאו יותר מאוחר ה-chips הללו הוחלפו ב-chips אחרים. בכלל, עם הזמן היה איחוד של Support Chips מסוימים, החלפתם באחרים ואפילו שילובם בתוך ה-CPU. אבל הפונקציונליות של ה-ports נשמר בדרך כלל, משיקולי "תאימות לאחור". לכן המודל שהיה ל-8086 רלוונטי במידה מסוימת גם היום.

לפיכך ב-8086 היו ה-CPU, שני ה-chips הנוספים, ה-8259A וה-8255 והמקלדת מחוברים בצורה הבאה:



שימו לב שרק בחיבורים ל-CPU יש מספרי port.
 אלו כמובן לא כל החיבורים שיש: רק אילו שמעורבים בפסיקה 9. לדוגמא,
 ישנו port 21h בין ה-CPU ל-8259A המאפשר ל-CPU להשפיע על תפקוד ה-8259A -
 למשל להשבית פסיקות ספציפיות כמו פסיקה 8 או פסיקה 9.

התוכנית הזו קוראת מקשים מהמקלדת ומדפיסה אותם, אבל ללא שום סיוע לא ממערכת ההפעלה ולא מה-BIOS, אלא קוראת את הנתונים ישירות מהחומרה. במושגים מקצועיים התוכנית הזו מכילה device driver משלה למקלדת. בתוכנית מופיעים מספר בלוקים שנחשבים לעיקר התוכנית.

התוכנית מקבלת מהחומרה את ה-scan code של המקשים שנלחצים ומתמירה את הנתונים ל-Ascii code בעצמה. היא עושה את ההמרה תמיד לאותיות גדולות (Upper Case). היא מתעלמת מהמקשים שאין להם קוד Ascii וגם למקשי ה-Shift וה-Caps Lock לא יהיה שום ההשפעה עליה.

ההמרה נעשית באמצעות הטבלה ScanTable שבו עבור המקשים שיש להם קודי Ascii ה- i- i בטבלה הוא קוד ה-Ascii של המקש שה-scan code שלו הוא i. שים לב שלטבלה יש את מבנה העקרוני של הפריסה של מקשים על המקלדת. מקשים שאין להם קודי Ascii הערך בטבלה הוא אפס.

ההמרה עצמה נעשית על ידי פקודת המכונה XTALB שנקראית Translate byte פקודה ללא אופרנדים שנועדה להמרות מסוג זה. הפקודה מחשבת את הכתובת BX+AL ואת הבית שהיא מוצאת שם היא מציבה חזרה לתוך AL. אם תרצו משהו כמו

תוכן $AL = [BX+AL]$

במילים אחרות אם BX מצביע לטבלת המרה מהסוג הזה, אז XLATB תבצע מעין תרגום של AL דרך הטבלה. בתוכנית עצמה בבלוק IV מופיע קטע קוד

```
MOV BX,OFFSET SscanTable
XLATB
```

התוכנית הראשית מבצעת השתלטות על פסיקה מספר 9 ולאחר מכן נכנסת ללולאה שבה היא ממתינה למידע על מקשים ומדפסה אותם עד שנודע לה שנלחץ המקש Esc. בבלוק I היא משתלטת על פסיקה 9 ובבלוק II היא משחזרת לפני חזרה ל-DOS. החזרה ל-DOS נעשה ע"י הדרך הישנה - הסתעפות ל-INT 20h בראש ה-PSP כפי שמתואר בתקציר מספר 10. הדפסה של התו נעשה ע"י קריירה לפרוצדורה DispChar הקוראת מצידה לפסיקת BIOS INT 10h אופציה AH = 14. התוכנית הראשית מקבלת ממטפל הפסיקה את תוכן המקשים באמצעות המשתנה גלובלי Buffer ומסמן שיש אינפורמציה חדשה באמצעות המשתנה הגלובלי Buff_Flag. כל זה נעשה בתוך הרוטינה Kbget הנקראית ע"י

התוכנית הראשית. אינני חושב שיש צורך מיוחד להתעקב על כל אלה משום שהם אינם מה שאנחנו רוצים להמחיש כאן.

מה שחשוב יותר הוא רוטינת הטיפול בפסיקה 9 (ה-device driver עצמו) הנקרא Kbint. כבלוק III הוא קורא את המקש ומאותת לבקר ההתקנים 8255 שהמקש נקרא.

בפקודה

IN AL,60h

הרוטינה קוראת מה-8255 את תוכן לחיצת המקש, כולל הביט המשמעותי ביותר המבדיל בין לחיצה (0) לשחרור (1) ומשמרת אותו מיד לאחר כך ע"י פקודת PUSH.

בפקודה

IN AL,61h

התוכנית קוראת מה-8255 את תוכן מספר port 61h ובפקודות

OR AL,80h
OUT 61h,AL
AND AL,7Fh
OUT 61h,AL

היא מודיעה ל-8255 על קריאת המקש ע"י כותבת את המספר שקראה מה-port חזרה לתוכו פעם עם הביט המשמעותי דלוק ולאחר מכן כאשר הוא כבוי.

כבלוק IV הרוטינה בודקת אם מדובר בלחיצה או שחרור ע"י הפקודה

TEST AL,80h

פקודת המכונה TEST מבצע AND לוגי על שני האופרנדים מבלי לשנות אותם - TEST היא גירסה של AND המשפיעה רק על אוגר הדגלים. ביצוע TEST של בית עם הקבוע 80h פירושו $ZF = 1$ אם הביט המשמעותי ביותר של הבית הוא אפס $ZF = 0$ אם הביט המשמעותי ביותר הוא 1. במקרה ש- $ZF = 1$ מדובר בשחרור מקש והתוכנית מתעלמת ממנו. אחרת היא מתמירה את הקוד ל-Ascii דרך הטבלה. אם מדובר במקש שאין לו קוד Ascii (תוצאת ההמרה היא אפס)

זו תהיה עוד מקרה שהתוכנית מתעלמת מהמקש. אחרת היא מציבה את תוצאת ההמרה במשתנה הגלובלי Buffer ומדליקה את משתנה הדגל הגלובלי Buff_Flag.

כבלוק V מתבצע ההודעה לבקר הפסיקות 8259A שהפסיקה טופלה. הדבר נעשה ע"י צמד הפקודות

```
MOV AL,20h
```

```
OUT 20h,AL
```

לאחר מכן הרוטינה משחזרת אוגרים ומבצעת IRET.

```

;
; keyb6.asm
; Example of custom keyboard support software
;
Stack1    SEGMENT PARA STACK 'STACK'
          DB 256 DUP(?)           ; 256d bytes of stack space
Stack1    ENDS
;
Data      SEGMENT PARA PUBLIC 'Data'
Buffer    DB ?                   ; 1 byte keyboard buffer
Buff_Flag DB ?                   ; 0 - Buffer empty, 1 - Char in
                                   ; Buffer
Msg3      DB ' PRESS ANY KEY AND YOU WILL SEE IT ON THE'
          DB ' SCREEN (PRESS ESC TO QUIT)',10,13,'$'
Oldbios_Off DW 0
Oldbios_Seg DW 0

; ScanTable converts scan codes received from the keyboard
; into their corresponding ascii character codes:
;
ScanTable DB 0,1,'1234567890-=',8,0
          DB 'QWERTYUIOP[]',0Dh,0
          DB 'ASDFGHJKL;',0,0,0,0
          DB 'ZXCVBNM,./',0,0,0
          DB ' ',0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
          DB '789-456+1230.'

Data      ENDS
;
Code      SEGMENT PARA PUBLIC 'Code'
Start     PROC FAR
;
; Standard program prologue
;
    ASSUME CS:Code
    PUSH DS      ; Save PSP SED addr
    MOV AX,0
    PUSH AX      ; Save RET addr offset (PSP+0)
    MOV AX,Data
    MOV DS,AX    ; Establish Data SEG addressability
    ASSUME DS:Data
;
; Part1: Setup our own keyboard interrupt service routine
;
    MOV AH,09
    MOV DX,OFFSET Msg3
    INT 21h
;

```

```

;=====
;                                     ###
;                                     #
;                                     #
;                                     ###
    CLI                                     ;
    MOV AX,0                               ;
    MOV ES,AX                               ; Point extra segment at the ...
;                                     ... interrupt service routine address table
    MOV DI,36                               ; Offset of entry for type code 09h
    MOV AX,ES:[DI]                           ;save old bios offset
    MOV Oldbios_Off,AX                       ;in Oldbios_Off
    MOV WORD PTR ES:[DI],OFFSET Kbint        ;
;                                     ; Offset of our service rout
    MOV AX,ES:[DI+2]                         ;save old bios segment
    MOV Oldbios_Seg,AX                       ;in Oldbios_Seg
    MOV ES:[DI+2],CS                         ; SEG of our service routine
    STI                                     ; Enable interrupts to the CPU
;
;=====
; Part2: Read from keyboard and display chars on screen
;
Forever:
    CALL Kbget                               ; Wait for a character from the keyboard
    CMP AL,01                               ; check for "esc"=quit =01
    JZ Quit
    PUSH AX                                  ; Save the character
    CALL DispChar                            ; Display the character received
    POP AX                                   ; Restore the character
    CMP AL,0Dh                               ; Was it carriage return?
    JNZ Forever                             ; Repeat loop if not
    MOV AL,0Ah                               ; Yes it was, we must also display ...
    CALL DispChar                            ; ... aline feed.
    JMP Forever                             ; Stay in this loop forever.
;
;=====
;                                     ### ###
;                                     #  #
;                                     #  #
;                                     ### ###
Quit:
    CLI                                     ;disable interrupt
    MOV AX,0                               ;restore the old bios keyboard routine
    MOV ES,AX                               ;
    MOV DI,36                               ;
    MOV AX,Oldbios_Off                     ;
    MOV ES:[DI],AX                         ;
    MOV AX,Oldbios_Seg                     ;
    MOV ES:[DI+2],AX                       ;
    STI                                     ;
;
; Return to DOS
;
    RET                                     ;
;=====

```

```

;
; Call Kbget to wait for a character to be received from
; the keyboard. The character is returned in reg AL.
;
Kbget PROC NEAR
    CLI ; Disable interrupts
    CMP Buff_Flag,0 ; Is buffer empty?
    JNZ Kbget2 ; ->No
    STI ; Re-enable interrupts
    JMP Kbget ; Wait until something in buffer
; There is something in the buffer, get it:

```

```

Kbget2:
    MOV AL,Buffer ; Get char at buffer start
    MOV Buff_Flag,0 ; Signal Buffer empty
    STI ; Re-enable interrupts
    RET ; Return from Kbget

```

```

Kbget ENDP

```

```

;
; Kbint is our own interrupt service routine
;

```

```

Kbint PROC FAR
    PUSH DS ; save all altered registers
    PUSH BX
    PUSH AX

```

```

;
; Establish addressability of our data segment:
;

```

```

    MOV AX,Data
    MOV DS,AX

```

```

;
;=====
;          ###  ###  ###
;          #   #   #
;          #   #   #
;          ###  ###  ###
;
; Read the keyboard data and send acknowledge signal
;
    IN  AL,60h ; Read keyboard input
    PUSH AX ; Save keyboard input
    IN  AL,61h ; Read 8255 port pb
    OR  AL,80h ; Set keyboard acknowledge signal
    OUT 61h,AL ; Send keyboard acknowledge signal
    AND AL,7Fh ; Reset keyboard acknowledge signal
    OUT 61h,AL ; Restore original 8255 port pb
;
;=====
;

```

```

;=====
;                                     ### # #
;                                     # # #
;                                     # # #
;                                     ### #
;
; Decode the scan code received:
;
    POP     AX                      ; Regain the keyboard input (AL)
    TEST    AL,80h                  ; Is it a key being released
    JNZ     Kbint2                  ; Branch if yes, we ignore these
    MOV     BX,OFFSET ScanTable    ; Scan code - ASCII table
    XLATB                                ; Convert the scan code to
                                ; an ASCII char
    CMP     AL,0                    ; Is it a valid ASCII key
    JZ      Kbint2                  ; Branch if not
;
; Place the ASCII character into the buffer:
;
    MOV     Buffer,AL                ; Place char in buffer
    MOV     Buff_Flag,1             ; Signal char in buffer
;
;=====
;
;
;=====
;                                     # #
;                                     # #
;                                     # #
;                                     #
;
; Now indicate "END OF INTERRUPT" to the interrupt controller:
;
Kbint2:
    MOV     AL,20h                  ; Send "EOI" command ...
    OUT     20h,AL                  ; ... to 8259 command register
;
;=====
;
    POP     AX                      ; Restore all altered registers
    POP     BX
    POP     DS
    IRET                             ; Return from interrupt
Kbint ENDP
;
; Subroutine to display a character on the screen.
; Enter with AL = character to be displayed
; Uses video interface in BIOS
;
DispChar PROC NEAR
    PUSH    BX                      ; Save BX register
    MOV     BX,0                    ; Select display page 0
    MOV     AH,14                   ; Function code for write
    INT     10h                     ; Call video driver in BIOS
    POP     BX                      ; Restore BX register
    RET                                     ; Return to caller of 'DispChar'
DispChar ENDP
;
Start ENDP
Code ENDS
END Start

```

367

E:\>keyb6.exe

PRESS ANY KEY AND YOU WILL SEE IT ON THE SCREEN (PRESS ESC TO QUIT)
USER ECHO
IS ALWAYS UPPER CASE

E:\>